Mahasagar-Bulletin of the National Institute of Oceanography, 19 (2), 1986, 123-128

A GRAPHICS DUMP ROUTINE FOR THE HP 9845B COMPUTER

E. S. DE SA

National Institute of Oceanography Dona Paula, Goa-403 004.

ABSTRACT

The note describes a software program written in BASIC, that permits graphical information on the CRT of a computer to be sent via an RS-232-C serial link to a graphics dot matrix printer for hardcopy outputs. As computers, printers and interface cards differ in most respects from manufacturer to manufacturer, the example chosen here is written for the HP 9845B desk top computer interfaced to a Paper Tiger Model IDS 440 dot matrix printer via an HP 98036 Rs-232C interface card.

Key-words: Software, Graphics, Computer.

Graphics dump software which is usually sold with most microcomputer systems serves the purpose of transferring any graphical data (e.g. graphs, histograms, engineering drawings etc.) from the CRT screen to a suitable printer with a graphics option eg. several brands of dot matrix printers in the market. The manufacturer normally supplies the required software on a diskette or burns it on a ROM (Read Only Memory) resides on the printer interface card. As is usually the case, the software will execute correctly on the brand of printer chosen by the user, or suggested by the manufacturer. Furthermore such software will be written in assembly language and if resident on ROMs will be virtually inaccessible to the curious user for modification.

The motivation for writing a graphics dump routine made itself clear when using the HP 9845B desk top computer to obtain graphical plots on the small sensitive paper and is not suitable for lengthy printing tasks. It was felt, therefore that a low cost option using a standard printer such as the IDS-440 Paper Tiger would satisfy the requirements of easy serviceability, with text and graphics printed on readily available computer brand paper.

Hardware Description

HP 9845B Computer

The HP 9845B uses a bit mapped graphics scheme to display images on its CRT screen. Thus a binary bit set to one in the display file results in a pixel dot being lit on the screen, and vice versa. The screen may be thought of as a matrix array consisting of 36 columns and 455 horizontal rows (in actual use, the 36th column is required but not displayed). Therefore each row contains 560 pixel dots, such that each column is 16 bits (or dots) wide. A glance at fig. 1 shows each



Fig. 1 Screen matrix contents

array element represented as a 16 bit binary word with MSB (most significant bit) bit left justified . It can be shown that the subscript K of an element A (K) is related to the row and column number of the screen matrix by :

$$K = 36^{*}\ \text{Row} + \text{Col.} \quad \ldots\ldots\ldots (1)$$

where the variables Row and Col. lie within the ranges $0 \leq \text{Row} \leq 454$ and $1 \leq \text{Col} \leq 36$ respectively. The validity of equation (1) can be checked by substituting values of the variables Row and Col. The matrix contains a total of (36^{*} $455 + 1 = 16381$) elements. The extra element is used as a pointer to the loading or storage of a graphics image. The range of values for the pointer is from -1 to -16379, and must be a negative number (Graphics, ROM Manual HP 9845) By assigning a value of -1, the BASIC statement GSTORE A (*) will store the entire 32K CRT memory contents in the array A. Likewise, it is possible to store one half of the graphics display (227 dot rows) by letting the pointer equal -8173 (See equation 1) Having transferred the CRT image into the integer array A, it is now possible to access the graphics data for transmission to a printer.

IDS-440 Printer

This low cost printer is capable of both serial and parallel modes of operation besides having graphics capability in both modes. The graphics mode of operation is enabled when the printer receives an ascii ETX (hex 03) control character while in normal print mode. The graphics option also expands the printer buffer to 2048 bytes.

The printer employs a raster scan technique which enables the printing of vertical columns of upto 7 dots across a page during each pass of the print head (fig. 2). The seven dot positions in any vertical column represents the binary bit



Fig. 2 Illustration on contiguous scanning
(taken from IDS-440 Owner's manual)

configuration of one received data character (IDS-440 Matrix Printer Owners Mannual).

When plotting continuous dot patterns vertically down a page via contiguous horizontal scanning of the print head, bit 6 in each received character is overwritten by bit 0 from each character during the following scan as shown in fig 2. As a result only 6 bits in a vertical dot column are available for plotting during contiguous scanning wih bit 6 set to zero in each character sent to the printer.

Flow Chart Description

The flowchart shown in fig. 3 begins by storing the CRT screen contents into an integer array using the GSTORE statement. The pointer value — 1 is set equal to A(0). The initialisation of the Intel 8251 USART chip on the HP-98036 serial card programs the card for asynchronous communication with a character length of 7 bits, no parity check, and 2 stop bits. The graphics mode of the printer is switched on by sending an ETX character to the printer using the WRITE BIN statement (line 190 in list out).

The actual dump of graphical characters commences by selecting the top seven rows of the screen. (Fig. 1). Working from column 1 to column 35 across the screen (i.e. top left to the top right corner) the MSB bits starting from row 6 and ending at row 0 are isolated from the 16 bit array elements associated with



Fig. 3 Graphics dump flowchart

these rows. These MSB bits are assembled to form a 7 bit character which is transmitted to the printer as a vertical column of 6 bits (or dots). The Data set Ready line is checked after each character is sent. Subsequent vertical dot columns are sent by selecting the next lower set of MSB bits from the 16 bit words in row 6 to 0 of column 1. The process is repeated for all bits in a column, and then for all 35 columns.

When a complete horizontal scan covering a block of 7 rows is complete, a vertical tab (ascii 11) must be sent to the printer indicating that a complete 7 row scan line of 560 bytes is over. The printer maintains a counter with the number of lines that are placed in memory. Whenever that count exceeds zero, the

```
10    ! ********************************************************
20    ! PROGRAM DPRINT DUMPS CRT GRAPHICS TO PAPER TIGER PRINTER
30    ! ********************************************************
40    PLOTTER IS 13,"GRAPHICS"
50    GRAPHICS
60    OPTION BASE 0
70    INTEGER Array(16381)
80    Array(0)=-1
90    GSTORE Array(*)                    !GSTORE stores pattern in R/W mem.
100   ! -----------------------------------
110   ! Prepare RS-232 interface card
120   ! -----------------------------------
130   WAIT WRITE 6,5;32                  !Interface Reset
140   WAIT WRITE 6,5;1                   !Control Mode
150   WRITE BIN 6;80                     !Usart Reset
160   WRITE BIN 6;203                    !Usart Mode Word 7 BIT char lenght
170   WRITE BIN 6;37                     !New Usart Control
180   WAIT WRITE 6,5;0                   !Data mode
190   WRITE BIN 6;3                      !Switch on printer graphics mode
200   ! -----------------------------------
210   ! Now do the actual dump of CRT graphics page
220   ! -----------------------------------
230   ! Initialize variables
240   J=0
250   I=6
260   X=1
270   Y=0
271   ! ---------------------------------------
290   WHILE I<=454                       ! While Loop starts here
300     FOR Col=1 TO 35                  ! Column Loop
310       FOR Bit=1 TO 16                ! Bit Loop
320         Y=0                          ! Clear Y for next char transmission
330         FOR Row=I TO J STEP -1       ! Row Loop
340           K=36*Row+Col-1             ! Array address equation
360           Cons=Array(K)              ! Select an array element
370           Cons=ROTATE(Cons,-Bit)     ! Move selected bit into LSB position
380           Andx=BINAND(Cons,X)        ! Isolate by ANDING with var-X
390           Y=BINIOR(Andx,Y)           ! Inclusive OR
400           Y=SHIFT(Y,-1)              ! Left shift by 1 bit position
410         NEXT Row
420         Y=SHIFT(Y,1)                 ! Adjust 6bit word by right shift
430         Y=BINAND(Y,63)               ! Bit 6=0
440   ! -----------------------------------
450   ! Check DSR Line of the printer
460   ! -----------------------------------
470         WAIT WRITE 6,5;1             ! Control mode
480 Loop:   R=READBIN(6)                 ! Read Usart Status
490         DISP "STATUS=";R             ! Display status.
500         S=BINAND(R,128)              ! Mask for DSR bit
510         IF S=0 THEN Loop
520         WAIT WRITE 6,5;0             ! Back to data mode
530   ! -----------------------------------
540         WRITE BIN 6;Y                ! Transmit the char.
550       NEXT Bit
560     NEXT Col                         ! Go for next MSB bit
570     WRITE BIN 6;3,11                 ! Send a VTAB char
580     I=I+6                            ! Select next 7 rows down
590     J=J+6
600   END WHILE
610   ! ---------------------------------------
620   WRITE BIN 6;3,2                    ! Place printer in text mode
630   WAIT WRITE 6,5;1                   ! Reset communications channel
640   WRITE BIN 6;64
650   WRITE BIN 6;74
660   WAIT WRITE 6,5;0
670   PRINT "************ DUMP GRAPHICS OVER **************"
680   END
690   ! ---------------------------------------
```

Fig. 4

multi tasking software of the printer takes characters out of its memory buffer and begins printing.

Subsequent printing lines are sent by selecting at a time a block of 7 rows e.g. row 12 to row 6 etc. as shown in the flowchart. When all 455 rows are printed, the dump operation ends, and the printer is placed in text mode for program listouts and other numeric outputs.

Use of the HP9845B binary I/O statements (see lines 370 onwards in fig. 4). simplifies the method of selecting successive MSB bits from array elements in different rows for the assembly of a data character. The program lines 370 to 430 of Fig. 4. uses the ROTATE statement to move a specific bit of the array element into an LSB position of the character under assembly. The BINAND, BINIOR and SHIFT statements (I/O ROM Manual HP 9845.) isolate and move the selected bits from a block of rows for a given column before data transmission to the printer. When the graphics buffer of the printer is full, it drops the DSR bit of the USART status word in the HP 98036 card. This line is monitored in the program so that data in the print buffer is not garbled, especially in situations where high baud rates (1200 baud) are selected by the user.

The simple BASIC program described here demonstrates one of several techniques used in writing graphics dump routines.

REFERENCES

Graphics Rom (Part No. 09845-91051). 1981. Hewlett Packard Desktop Computer Division, Colorado

IDS-440 Impact Printer (Part No. 9000-000001). 1979. Integral Data System Inc. Massachusetts

I/O ROM (Part No. 09845-93060). 1981. Hewlett Packard Desktop Computer Division, Colorado